In this Maple worksheet, we study linear elasticity in a space of dimension 2. For more details, we refer to J.-F. Pommaret, *Partial Differential Control Theory*, Kluwer, 2001, and L. Landau, E. Lifschitz, *Physique théorique*, Tome 7: *Elasticité*, second edition, MIR, 1990.

```
> with(Ore_algebra):
> with(OreModules):
```

We define the Weyl algebra $Alg = A_2$, where D_i represents the differential operator w.r.t. x_i, $i = 1, 2$.

```
> Alg := DefineOreAlgebra(diff=[D[1],x[1]], diff=[D[2],x[2]],
> polynom=[x[1],x[2]], comm=[alpha,beta]):
```

Using the Lie derivative of the euclidean metric $w[i, j] = \delta[i, j]$, where $\delta[i, j]$ denotes the Kronecker symbol, we obtain the following *Killing operator*:

```
> R := evalm([[D[1],0],[D[2]/2,D[1]/2],[D[2]/2,D[1]/2],[0,D[2]]]);
```

$$R := \begin{bmatrix} \mathrm{D}_1 & 0 \\ \frac{1}{2}\mathrm{D}_2 & \frac{1}{2}\mathrm{D}_1 \\ \frac{1}{2}\mathrm{D}_2 & \frac{1}{2}\mathrm{D}_1 \\ 0 & \mathrm{D}_2 \end{bmatrix}$$

The symmetric small strain tensor ε is defined by

```
> ApplyMatrix(R, [seq(xi[i](x[1],x[2]),i=1..2)],Alg)=
> evalm([[epsilon[1,1](x[1],x[2])],[epsilon[1,2](x[1],x[2])],
> [epsilon[2,1](x[1],x[2])],[epsilon[2,2](x[1],x[2])]]);
```

$$\begin{bmatrix} \frac{\partial}{\partial x_1}\xi_1(x_1, x_2) \\ \frac{1}{2}\left(\frac{\partial}{\partial x_2}\xi_1(x_1, x_2)\right) + \frac{1}{2}\left(\frac{\partial}{\partial x_1}\xi_2(x_1, x_2)\right) \\ \frac{1}{2}\left(\frac{\partial}{\partial x_2}\xi_1(x_1, x_2)\right) + \frac{1}{2}\left(\frac{\partial}{\partial x_1}\xi_2(x_1, x_2)\right) \\ \frac{\partial}{\partial x_2}\xi_2(x_1, x_2) \end{bmatrix} = \begin{bmatrix} \varepsilon_{1,1}(x_1, x_2) \\ \varepsilon_{1,2}(x_1, x_2) \\ \varepsilon_{2,1}(x_1, x_2) \\ \varepsilon_{2,2}(x_1, x_2) \end{bmatrix}$$

where $\xi = (\xi[1], \xi[2])^T$ denotes the displacement. The rank of the *Alg*-module M associated with R is:

```
> OreRank(R, Alg);
```

$$0$$

Therefore, the *Alg*-module M is torsion. Let us compute the autonomous elements of the system:

```
> AutonomousElements(R, [xi[1](x[1],x[2]),xi[2](x[1],x[2])], Alg);
```

$$\left[\begin{bmatrix} \left(\frac{\partial}{\partial x_2}\theta_1(x_1, x_2)\right) + \left(\frac{\partial}{\partial x_1}\theta_2(x_1, x_2)\right) = 0 \\ \frac{\partial}{\partial x_1}\theta_1(x_1, x_2) = 0 \\ \frac{\partial^2}{\partial x_2^2}\theta_1(x_1, x_2) = 0 \\ \frac{\partial}{\partial x_2}\theta_2(x_1, x_2) = 0 \end{bmatrix}, \begin{bmatrix} \theta_1 = \_C1\, x_2 + \_C2 \\ \theta_2 = -\_C1\, x_1 + \_C3 \end{bmatrix}, \begin{bmatrix} \theta_1 = \xi_1(x_1, x_2) \\ \theta_2 = \xi_2(x_1, x_2) \end{bmatrix}\right]$$

Hence, $R\,\xi = 0$ implies that $\xi_1 = C1\ x2 + C2$ and $\xi_2 = -C1\ x1 + C3$, where $C1$, $C2$ and $C3$ are three constants. The displacement ξ which satisfies $R\,\xi = 0$ generates a Lie group of transformations with the three infinitesimal generators D_1, D_2 and $x_1\,\mathrm{D}_2 - x_2\,\mathrm{D}_1$ (two translations and one rotation).

The compatibility conditions of the strain tensor ε are defined by

```
> R2 := SyzygyModule(R, Alg);
```

$$R2 := \begin{bmatrix} D_2{}^2 & 0 & -2\,D_1\,D_2 & D_1{}^2 \\ 0 & 1 & -1 & 0 \end{bmatrix}$$

or equivalently by:

```
> ApplyMatrix(R2, [epsilon[1,1](x[1],x[2]),epsilon[1,2](x[1],x[2]),
> epsilon[2,1](x[1],x[2]),epsilon[2,2](x[1],x[2])], Alg)=evalm([[0]$2]);
```

$$\begin{bmatrix} (\frac{\partial^2}{\partial x_2{}^2}\,\varepsilon_{1,\,1}(x_1,\,x_2)) - 2\,(\frac{\partial^2}{\partial x_2\,\partial x_1}\,\varepsilon_{2,\,1}(x_1,\,x_2)) + (\frac{\partial^2}{\partial x_1{}^2}\,\varepsilon_{2,\,2}(x_1,\,x_2)) \\ \varepsilon_{1,\,2}(x_1,\,x_2) - \varepsilon_{2,\,1}(x_1,\,x_2) \end{bmatrix} = \begin{bmatrix} 0 \\ 0 \end{bmatrix}$$

Let us point out that the compatibility conditions of the strain tensor ε are of order 2 whereas the Killing operator is only of order 1. This result comes from the fact that the Killing operator defines a formally integrable system which is not involutive. For more details, see J.-F. Pommaret, *Lie Pseudogroups and Mechanics*, Gordon and Breach, 1998.

We know that the compatibility conditions of the the strain tensor ε are parametrized by the Killing operator. Let us check this result by using extension modules. We first define the formal adjoint *R2_adj* of *R2*.

```
> R2_adj := Involution(R2, Alg);
```

$$R2\_adj := \begin{bmatrix} D_2{}^2 & 0 \\ 0 & 1 \\ -2\,D_1\,D_2 & -1 \\ D_1{}^2 & 0 \end{bmatrix}$$

Then, the compatibility conditions of the strain tensor ε are parametrizable if and only if the *Alg*-module associated with *R2* is torsion-free or, equivalently, if and only if the first extension module ext^1 with values in *Alg* of the *Alg*-module associated with *R2_adj* is 0.

```
> Ext1 := Exti(R2_adj, Alg, 1);
```

$$Ext1 := \left[\begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}, \begin{bmatrix} D_2{}^2 & 0 & -2\,D_1\,D_2 & D_1{}^2 \\ 0 & 1 & -1 & 0 \end{bmatrix}, \begin{bmatrix} -2\,D_1 & 0 \\ -D_2 & -D_1 \\ -D_2 & -D_1 \\ 0 & -2\,D_2 \end{bmatrix} \right]$$

As the first matrix *Ext1*[1] is the identity matrix, we obtain that the *Alg*-module associated with *R2* is torsion-free and the compatibility conditions are parametrized by *Ext1*[3]. We easily check that up to the constant $-\frac{1}{2}$, we recover the Killing operator. Let us check whether or not the *Alg*-module associated with *R2* is reflexive.

```
> Exti(R2_adj, Alg, 2);
```

$$\left[\begin{bmatrix} D_1 & 0 \\ D_2{}^2 & 0 \\ 0 & D_2 \\ 0 & D_1{}^2 \end{bmatrix}, \begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}, \mathrm{SURJ}(2) \right]$$

As the first matrix is not the identity matrix, we obtain that the *Alg*-module associated with *R2* is not reflexive, and thus, not free.

Let us define the formal adjoint *R_adj* of *R*.

```
> R_adj := Involution(R, Alg);
```

$$R\_adj := \begin{bmatrix} -\mathrm{D}_1 & -\frac{1}{2}\,\mathrm{D}_2 & -\frac{1}{2}\,\mathrm{D}_2 & 0 \\ 0 & -\frac{1}{2}\,\mathrm{D}_1 & -\frac{1}{2}\,\mathrm{D}_1 & -\mathrm{D}_2 \end{bmatrix}$$

If we consider -*R_adj* as a differential operator from the symmetric stress tensor σ to the density of forces f, then we have:

```
> ApplyMatrix(evalm(-R_adj), [sigma[1,1](x[1],x[2]),sigma[1,2](x[1],x[2]),
> sigma[1,2](x[1],x[2]),sigma[2,2](x[1],x[2])], Alg)=
> evalm([seq([f[i](x[1],x[2])],i=1..2)]);
```

$$\begin{bmatrix} (\frac{\partial}{\partial x_1}\,\sigma_{1,1}(x_1,\,x_2)) + (\frac{\partial}{\partial x_2}\,\sigma_{1,2}(x_1,\,x_2)) \\ (\frac{\partial}{\partial x_1}\,\sigma_{1,2}(x_1,\,x_2)) + (\frac{\partial}{\partial x_2}\,\sigma_{2,2}(x_1,\,x_2)) \end{bmatrix} = \begin{bmatrix} f_1(x_1,\,x_2) \\ f_2(x_1,\,x_2) \end{bmatrix}$$

Let us check whether or not there exist compatibility conditions for the previous inhomogeneous system:

```
> SyzygyModule(R_adj, Alg);
```

$$\mathrm{INJ}(2)$$

We obtain that there is no compatibility condition on the density of forces f.

Let us check if the stress tensor ε is parametrizable when $f[i] = 0$, $i = 1, 2$.

```
> ext1 := Exti(R, Alg, 1);
```

$$ext1 := \left[\begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}, \begin{bmatrix} 2\,\mathrm{D}_1 & \mathrm{D}_2 & \mathrm{D}_2 & 0 \\ 0 & \mathrm{D}_1 & \mathrm{D}_1 & 2\,\mathrm{D}_2 \end{bmatrix}, \begin{bmatrix} {\mathrm{D}_2}^2 & 0 \\ 0 & 1 \\ -2\,\mathrm{D}_1\,\mathrm{D}_2 & -1 \\ {\mathrm{D}_1}^2 & 0 \end{bmatrix} \right]$$

Therefore, we obtain that the stress tensor σ is then parametrizable and a parametrization is defined by *ext*[3].

```
> ApplyMatrix(ext1[3], [seq([lambda[i](x[1],x[2])],i=1..2)],Alg)=
> evalm([[sigma[1,1](x[1],x[2])],[sigma[1,2](x[1],x[2])],[sigma[2,1](x[1],x[2])],
> [sigma[2,2](x[1],x[2])]]);
```

$$\begin{bmatrix} \frac{\partial^2}{\partial {x_2}^2}\,\lambda_1(x_1,\,x_2) \\ \lambda_2(x_1,\,x_2) \\ -2\,(\frac{\partial^2}{\partial x_2\,\partial x_1}\,\lambda_1(x_1,\,x_2)) - \lambda_2(x_1,\,x_2) \\ \frac{\partial^2}{\partial {x_1}^2}\,\lambda_1(x_1,\,x_2) \end{bmatrix} = \begin{bmatrix} \sigma_{1,1}(x_1,\,x_2) \\ \sigma_{1,2}(x_1,\,x_2) \\ \sigma_{2,1}(x_1,\,x_2) \\ \sigma_{2,2}(x_1,\,x_2) \end{bmatrix}$$

Using the fact that the stress tensor σ is symmetric, i.e., $\sigma[1{,}2] = \sigma[2{,}1]$, we obtain that $\lambda_2 = -\mathrm{D}_1\,\mathrm{D}_2\,\lambda_1$. Hence, if we neglect the dependent variable $\sigma[2{,}1]$, then we obtain the following parametrization:

```
> ApplyMatrix(linalg[submatrix](ext1[3], [1,3,4], 1..1),
> [seq(lambda[i](x[1],x[2]),i=1..2)], Alg)
> =evalm([[sigma[1,1]],[sigma[1,2]],[sigma[2,2]]]);
```

$$\begin{bmatrix} \frac{\partial^2}{\partial {x_2}^2}\,\lambda_1(x_1,\,x_2) \\ -2\,(\frac{\partial^2}{\partial x_2\,\partial x_1}\,\lambda_1(x_1,\,x_2)) \\ \frac{\partial^2}{\partial {x_1}^2}\,\lambda_1(x_1,\,x_2) \end{bmatrix} = \begin{bmatrix} \sigma_{1,1} \\ \sigma_{1,2} \\ \sigma_{2,2} \end{bmatrix}$$

In the literature, the function λ_1 is usually called the *Airy function.*

Now, let us consider the constitutive law, i.e., the *Hooke law* defined by the matrix

```
> B := evalm([[alpha+2*beta,0,0,alpha],[0,2*beta,0,0],[0,0,2*beta,0],
> [alpha,0,0,alpha+2*beta]]);
```

$$B := \begin{bmatrix} \alpha+2\,\beta & 0 & 0 & \alpha \\ 0 & 2\,\beta & 0 & 0 \\ 0 & 0 & 2\,\beta & 0 \\ \alpha & 0 & 0 & \alpha+2\,\beta \end{bmatrix}$$

where α and β are the *Lamé constants.* Then, the stress tensor σ is related to the strain tensor ε as follows:

```
> evalm([[sigma[1,1](x[1],x[2])],[sigma[1,2](x[1],x[2])],[sigma[2,1](x[1],x[2])],
> [sigma[2,2](x[1],x[2])]])=ApplyMatrix(B, [epsilon[1,1](x[1],x[2]),
> epsilon[1,2](x[1],x[2]),epsilon[2,1](x[1],x[2]),epsilon[2,2](x[1],x[2])], Alg);
```

$$\begin{bmatrix} \sigma_{1,1}(x_1,\,x_2) \\ \sigma_{1,2}(x_1,\,x_2) \\ \sigma_{2,1}(x_1,\,x_2) \\ \sigma_{2,2}(x_1,\,x_2) \end{bmatrix} = \begin{bmatrix} (\alpha+2\,\beta)\,\varepsilon_{1,1}(x_1,\,x_2)+\alpha\,\varepsilon_{2,2}(x_1,\,x_2) \\ 2\,\beta\,\varepsilon_{1,2}(x_1,\,x_2) \\ 2\,\beta\,\varepsilon_{2,1}(x_1,\,x_2) \\ \alpha\,\varepsilon_{1,1}(x_1,\,x_2)+(\alpha+2\,\beta)\,\varepsilon_{2,2}(x_1,\,x_2) \end{bmatrix}$$

Now, if we compose the parametrization $\varepsilon = R\,\xi$ of the strain tensor ε by the displacement ξ, the Hooke law $\sigma = B\,\varepsilon$ and the differential operator -$R\_adj\,\sigma = f$, then we obtain $T\,\xi = 0$, where $T = -R\_adj \circ B \circ R$ is defined by:

```
> T := Mult(-1, R_adj, B, R, Alg);
```

$$T := \begin{bmatrix} \mathrm{D}_1{}^2\,\alpha+2\,\mathrm{D}_1{}^2\,\beta+\mathrm{D}_2{}^2\,\beta & \mathrm{D}_1\,\alpha\,\mathrm{D}_2+\mathrm{D}_2\,\beta\,\mathrm{D}_1 \\ \mathrm{D}_1\,\alpha\,\mathrm{D}_2+\mathrm{D}_2\,\beta\,\mathrm{D}_1 & \mathrm{D}_1{}^2\,\beta+\mathrm{D}_2{}^2\,\alpha+2\,\mathrm{D}_2{}^2\,\beta \end{bmatrix}$$

In terms of ξ, we have the following equations

```
> ApplyMatrix(T, [seq(xi[i](x[1],x[2]),i=1..2)], Alg)=
> evalm([seq([f[i](x[1],x[2])],i=1..2)]);
```

$$\begin{bmatrix} (\alpha+2\,\beta)\,(\frac{\partial^2}{\partial x_1{}^2}\,\xi_1(x_1,\,x_2))+\beta\,(\frac{\partial^2}{\partial x_2{}^2}\,\xi_1(x_1,\,x_2))+(\alpha+\beta)\,(\frac{\partial^2}{\partial x_2\,\partial x_1}\,\xi_2(x_1,\,x_2)) \\ (\alpha+\beta)\,(\frac{\partial^2}{\partial x_2\,\partial x_1}\,\xi_1(x_1,\,x_2))+\beta\,(\frac{\partial^2}{\partial x_1{}^2}\,\xi_2(x_1,\,x_2))+(\alpha+2\,\beta)\,(\frac{\partial^2}{\partial x_2{}^2}\,\xi_2(x_1,\,x_2)) \end{bmatrix} = \begin{bmatrix} f_1(x_1,\,x_2) \\ f_2(x_1,\,x_2) \end{bmatrix}$$

called the *Navier equations.*

Let us point out that the Hooke law is invertible and its inverse is defined by:

```
> B_inv := linalg[inverse](B);
```

$$B\_inv := \begin{bmatrix} \frac{\alpha+2\,\beta}{4\,\beta\,(\alpha+\beta)} & 0 & 0 & -\frac{\alpha}{4\,\beta\,(\alpha+\beta)} \\ 0 & \frac{1}{2\,\beta} & 0 & 0 \\ 0 & 0 & \frac{1}{2\,\beta} & 0 \\ -\frac{\alpha}{4\,\beta\,(\alpha+\beta)} & 0 & 0 & \frac{\alpha+2\,\beta}{4\,\beta\,(\alpha+\beta)} \end{bmatrix}$$

Therefore, the strain tensor ε is related to the stress tensor σ as follows:

```
> evalm([[epsilon[1,1](x[1],x[2])],[epsilon[1,2](x[1],x[2])],
> [epsilon[2,1](x[1],x[2])],[epsilon[2,2](x[1],x[2])]])=ApplyMatrix(B_inv,
> [sigma[1,1](x[1],x[2]),sigma[1,2](x[1],x[2]),sigma[2,1](x[1],x[2]),
> sigma[2,2](x[1],x[2])], Alg);
```

$$\begin{bmatrix} \varepsilon_{1,1}(x_1, x_2) \\ \varepsilon_{1,2}(x_1, x_2) \\ \varepsilon_{2,1}(x_1, x_2) \\ \varepsilon_{2,2}(x_1, x_2) \end{bmatrix} = \begin{bmatrix} \frac{1}{4}\frac{(\alpha+2\beta)\,\sigma_{1,1}(x_1, x_2)}{\beta\,(\alpha+\beta)} - \frac{1}{4}\frac{\alpha\,\sigma_{2,2}(x_1, x_2)}{\beta\,(\alpha+\beta)} \\ \frac{1}{2}\frac{\sigma_{1,2}(x_1, x_2)}{\beta} \\ \frac{1}{2}\frac{\sigma_{2,1}(x_1, x_2)}{\beta} \\ -\frac{1}{4}\frac{\alpha\,\sigma_{1,1}(x_1, x_2)}{\beta\,(\alpha+\beta)} + \frac{1}{4}\frac{(\alpha+2\beta)\,\sigma_{2,2}(x_1, x_2)}{\beta\,(\alpha+\beta)} \end{bmatrix}$$

If we compose the parametrization $R2\_adj\,\lambda = \sigma$ of the stress tensor σ by the Airy functions λ, the inverse of the Hooke law $\varepsilon = B\_inv\,\sigma$ and the compability condition $R2\,\varepsilon = 0$ of the strain tensor ε, then we obtain $C\,\lambda = 0$, where C is defined by:

```
> C := simplify(Mult(R2, B_inv, R2_adj, Alg));
```

$$C := \begin{bmatrix} \frac{1}{4}\frac{\mathrm{D}_2{}^4\,\alpha + 2\,\mathrm{D}_2{}^4\,\beta + 6\,\mathrm{D}_2{}^2\,\mathrm{D}_1{}^2\,\alpha + 8\,\mathrm{D}_2{}^2\,\mathrm{D}_1{}^2\,\beta + \mathrm{D}_1{}^4\,\alpha + 2\,\mathrm{D}_1{}^4\,\beta}{\beta\,(\alpha+\beta)} & \frac{\mathrm{D}_1\,\mathrm{D}_2}{\beta} \\ \frac{\mathrm{D}_1\,\mathrm{D}_2}{\beta} & \frac{1}{\beta} \end{bmatrix}$$

Equivalently, we have the following system:

```
> ApplyMatrix(C, [seq(lambda[i](x[1],x[2]),i=1..2)], Alg)=evalm([[0]$2]);
```

$$\begin{bmatrix} \frac{1}{4}\frac{(6\,\alpha+8\,\beta)\,(\frac{\partial^4}{\partial x_2{}^2\,\partial x_1{}^2}\,\%1)}{\beta\,(\alpha+\beta)} + \frac{1}{4}\frac{(\alpha+2\,\beta)\,(\frac{\partial^4}{\partial x_2{}^4}\,\%1)}{\beta\,(\alpha+\beta)} + \frac{1}{4}\frac{(\alpha+2\,\beta)\,(\frac{\partial^4}{\partial x_1{}^4}\,\%1)}{\beta\,(\alpha+\beta)} + \frac{\frac{\partial^2}{\partial x_2\,\partial x_1}\,\lambda_2(x_1, x_2)}{\beta} \\ \frac{\frac{\partial^2}{\partial x_2\,\partial x_1}\,\%1}{\beta} + \frac{\lambda_2(x_1, x_2)}{\beta} \end{bmatrix} = \begin{bmatrix} 0 \\ 0 \end{bmatrix}$$

$$\%1 := \lambda_1(x_1, x_2)$$

From the second equation, we can express λ_2 in terms of λ_1 and, by substitution in the first one, we finally obtain the following system:

```
> E := Elimination(C, [lambda[1],lambda[2]], [0,0], Alg, [lambda[1]]):
> ApplyMatrix(E[1], [lambda[2](x[1],x[2])], Alg)=ApplyMatrix(E[2],
> [lambda[1](x[1],x[2])], Alg);
```

$$\begin{bmatrix} 0 \\ \lambda_2(x_1, x_2) \end{bmatrix} = \begin{bmatrix} -(\frac{\partial^4}{\partial x_1{}^4}\,\%1) - (\frac{\partial^4}{\partial x_2{}^4}\,\%1) - 2\,(\frac{\partial^4}{\partial x_2{}^2\,\partial x_1{}^2}\,\%1) \\ -(\frac{\partial^2}{\partial x_2\,\partial x_1}\,\%1) \end{bmatrix}$$

$$\%1 := \lambda_1(x_1, x_2)$$

In particular, λ_1 satisfies the partial differential equation $BiLaplacian\,\lambda_1 = 0$, where BiLaplacian is defined by:

```
> BiLaplacian := factor(-E[2][1,1]);
```

$$BiLaplacian := (\mathrm{D}_1{}^2 + \mathrm{D}_2{}^2)^2$$

Moreover, from the system $C\,\lambda = 0$, we also obtain the following equations:

```
> F := Elimination(C, [lambda[1],lambda[2]], [0,0], Alg, [lambda[2]]):
> ApplyMatrix(F[1], [lambda[1](x[1],x[2])], Alg)=ApplyMatrix(F[2],
> [lambda[2](x[1],x[2])], Alg);
```

$$\begin{bmatrix} 0 \\ \frac{\partial^2}{\partial x_2\,\partial x_1}\,\%1 \\ (\frac{\partial^4}{\partial x_1{}^4}\,\%1) + (\frac{\partial^4}{\partial x_2{}^4}\,\%1) \\ \frac{\partial^5}{\partial x_2{}^5}\,\%1 \end{bmatrix} = \begin{bmatrix} -(\frac{\partial^4}{\partial x_1{}^4}\,\%2) - (\frac{\partial^4}{\partial x_2{}^4}\,\%2) - 2\,(\frac{\partial^4}{\partial x_2{}^2\,\partial x_1{}^2}\,\%2) \\ -\%2 \\ 2\,(\frac{\partial^2}{\partial x_2\,\partial x_1}\,\%2) \\ (\frac{\partial^3}{\partial x_1{}^3}\,\%2) + 2\,(\frac{\partial^3}{\partial x_2{}^2\,\partial x_1}\,\%2) \end{bmatrix}$$

$$\%1 := \lambda_1(x_1,\,x_2)$$
$$\%2 := \lambda_2(x_1,\,x_2)$$

In particular, the first equation shows that λ_2 also satisfies the equation $BiLaplacian\,\lambda_2 = 0$.

Now, if we consider the case without where the density of forces f is 0, then we have

$$R\_adj\,\sigma = 0 \iff \sigma = R2\_adj\,\lambda.$$

Therefore, $\lambda = (\lambda_1,\ \lambda_2)^T$ must satisfy $BiLaplacian\,\lambda = 0$, and thus, λ_1 and λ_2 are two biharmonic functions.